

Excel-Tipp: Genaueres Alter berechnen

Versionen: Excel 97, 2000 und 2002/XP

Altersberechnungen werfen immer wieder Probleme auf. Als geeignete Funktion bietet sich Funktion DATEDIF an, die folgende Syntax besitzt:

=DATEDIF(<Startdatum>;<Enddatum>;"<Einheit>")

Damit die Funktion keinen Fehler erzeugt, muss das Startdatum vor dem Enddatum liegen. Für Geburtstage könnte das zum Beispiel bedeuten, dass als «Startdatum» der Geburtstag und als «Enddatum» die Tabellenfunktion HEUTE() eingesetzt wird. Bei der «Einheit», die als Text in Anführungszeichen eingegeben wird, kann es sich um folgende Werte handeln:

y	vollständige Jahre
m	vollständige Monate
d	vollständige Tage
md	nur Tage ohne Berücksichtigung von Monaten und Jahren
ym	nur Monate ohne Berücksichtigung der Jahre
yd	nur Tage ohne Berücksichtigung der Jahreszahlen

Angenommen, Sie hätten das Geburtsdatum «3.6.1969» in Zelle A1 eingegeben. Nun möchten Sie in Zelle B1 das tatsächliche Alter berechnen. Dazu geben Sie folgende Formel ein:

=DATEDIF(A1;HEUTE();"y")

Ausgehend vom aktuellen Jahr 2003 ist das Ergebnis entweder 33, wenn «HEUTE» vor dem 3.6. liegt, oder 34, wenn das aktuelle Datum danach ist.

Falls eine Person bereits verstorben ist, können Sie mit DATEDIF natürlich auch das erreichte Alter berechnen. Ändern Sie die oben angegebene Formel dahingehend, dass anstelle der Funktion HEUTE() ein Verweis auf die Zelle mit dem Sterbedatum steht. Wenn dieses etwa in Zelle A2 vorliegt, müsste die Formel wie folgt aussehen:

=DATEDIF(A1;A2;"y")

Die Berechnungen funktionieren hervorragend, solange Sie mit Datumsangaben nach dem 1.1.1900 arbeiten. Gerade in der Ahnenforschung ist es aber nahe liegend, dass Geburts- und Sterbedaten in früheren Jahrhunderten liegen. Mit diesen

Werten kann Excel nicht rechnen, sodass nur ein kleiner Trick zum gewünschten Resultat führt.

Der Trick besteht darin, die Datumsangabe in das Jahr 1900 zu verlegen und dann die Jahresanzahl vor 1900 zum Ergebnis der DATEDIF-Funktion zu addieren. Am besten veranschaulicht das ein Beispiel:

Angenommen, Sie hätten in A1 dieses Mal als Geburtstag den «30.5.1826» eingegeben und in B1 wollten Sie ausrechnen, wie alt die Person heute wäre. Dazu geben Sie in B1 folgende Formel ein:

```
=WENN(ISTTEXT(A1);DATEDIF(DATWERT(ERSETZEN(A1;LÄNGE(A1)-3;4;1900));HEUTE());"y")  
+(1900-RECHTS(A1;4));DATEDIF(A1;HEUTE());"y"))
```

Die Formel ist so ausgelegt, dass sie das Alter auch für Datumswerte ab 1900 ermitteln kann. Deshalb beginnt Sie mit einer WENN-Abfrage, die prüft, ob das zu untersuchende Datum von Excel ausgewertet werden kann oder nicht:

ISTTEXT(A1)

Datumswerte vor 1900 trägt Excel als Textwerte in die Tabelle ein, was im Fall des Datums «30.5.1826» zutrifft. Somit wird der «Dann-Teil» der WENN-Abfrage ausgeführt. Darin setzen Sie das Geburtsdatum auf einen Tag im Jahr 1900, und zwar mit folgender Funktion:

```
ERSETZEN(A1;LÄNGE(A1)-3;4;1900)
```

Die Funktion geht von vierstelligen Jahreszahlen aus und ersetzt die letzten vier Zeichen des Datums durch «1900», sodass Sie im Beispiel die Zeichenfolge «30.5.1900» erhalten. Eine DATWERT-Funktion wandelt diesen Text in ein gültiges Datum um, das in der Berechnung der Jahresdifferenz eingesetzt werden kann:

```
DATEDIF(DATWERT(ERSETZEN(A1;LÄNGE(A1)-3;4;1900));HEUTE());"y")
```

Damit kennen Sie das Alter, wenn die Person im Jahr 1900 geboren wäre. Um das in diesem Fall korrekte Alter für das Geburtsjahr 1826 herauszufinden, müssen Sie noch die Jahre zwischen Geburtsdatum und dem Jahr 1900 addieren:

```
+(1900-RECHTS(A1;4))
```

Alles in allem erfahren Sie auf diese Weise, dass eine Person, die am 30.5.1826 geboren worden ist, im Jahr 2003 entweder 176 (wenn das aktuelle Datum vor dem 30.5. liegt) oder 177 (nach dem 30.5.) wäre.

Auf dieselbe Art und Weise können Sie natürlich auch das bis zum Tod erreichte Alter einer Person berechnen. Dafür setzen Sie die zuvor beschriebene Formel einmal

für das Geburtsdatum und einmal für das Sterbedatum ein und anschliessend subtrahieren Sie die beiden Ergebnisse.

www.ubach.ch